
Lubrificazione idrdinamica

1. Primo quesito, problema inverso. Abbiamo una coppia prismatica lubrificata -con meato a spessore linearmente decrescente- che verifica i seguenti parametri:

1.1)	

essendo il significato di ciascuno illustrato in figura. Sia inoltre assegnata la viscosità del lubrificante (alla temperatura di funzionamento):

1.12)	

Allora si chiede di determinare:

1.13)	il diagramma dei gradienti delle pressioni;
1.14)	il diagramma delle pressioni per via analitica;
1.15)	il diagramma delle pressioni per via numerica;
1.16)	la pressione massima;
1.17)	la forza di sostentamento  per via analitica;
1.18)	la forza di sostentamento  per via numerica;
1.19)	il punto di applicazione  di ;
1.20) 	il coefficiente di attrito mediato ;
1.21)   il diagramma delle velocità in , , .

2. Discussione analitica.  In un generico meato in condizioni stazionarie si dimostra per la portata (per unità di larghezza del meato) l’espressione

2.1)	 

Indicando con  il valore dell’altezza per la quale si annulla il gradiente della pressione si ha in particolare

2.2)	

che sostituito nella 2.1 porge

2.3)	

Poiché deve[footnoteRef:1] risultare  possiamo scrivere: [1:  In questa trattazione si pone uguale a zero la pressione esterna, per cui la funzione delle pressioni porge quella che viene detta ‘sovrapressione’, che è poi ciò che serve per calcolare la forza di sostentamento.] 

 


da cui si deduce per confronto

2.4)	

Nel caso particolare di un meato a spessore linearmente decrescente la funzione  si scrive poi

2.5)	

e dunque, sostituendo nella 2.4 e integrando (sono integrali elementari) si perviene alla espressione

2.6)	

che sostituita nella 2.5 fornisce la corrispondente ascissa

2.7)	

Sostituendo la 2.6 nella 2.3 abbiamo il gradiente delle pressioni

2.8)	

Per integrare il gradiente conviene ricondursi alla variabile , considerando che

2.9)	

Ciò posto la 2.8 si scrive

2.10)	

L’integrazione a questo punto è piuttosto semplice:





Si ottiene dunque la pressione

2.11)	

Introdotto poi il coefficiente caratteristico

2.12)	

si ha

2.13)	

Per il calcolo della forza di sostentamento integrando la 2.11 si ha








E dunque 

2.14)	

Introdotto poi il coefficiente caratteristico si ha

2.15)	

Calcolo ora, attraverso la 2.11, il punto di applicazione  della risultante delle forze di pressione tale per cui la sollecitazione ridotta sia equivalente a quella data.










Introducendo poi  abbiamo




Sostituendo la 2.14 si ha allora 









Quindi ho trovato

2.16)	

Per procedere con l’analisi e pervenire al coefficiente di attrito mediato abbiamo bisogno del campo delle velocità. Dalla teoria generale  dei meati lubrificati abbiamo

2.17)	 

che inserendo la 2.8 si scrive

2.18)	

Considerando la legge di Petroff   possiamo allora scrivere



ovvero
2.19)	

Gli sforzi che il fluido esercita sulla parete superiore allora si ottengono ponendo  e sono

2.20)	

Per ricavare la forza  che complessivamente il fluido esercita sulla parete mobile, per unità di larghezza, dobbiamo integrare sulla lunghezza del meato:





Sostituendo poi il valore 2.6 si conclude

2.21)	

Introducendo poi il parametro caratteristico  abbiamo

2.22)	

È immediato allora il calcolo del coefficiente di attrito mediato, che indichiamo , come rapporto  . In realtà però 




Dunque 

2.23)	

Posso tuttavia scrivere anche



 
Per cui si ha

2.24)	

avendo posto

2.25)	

Riassumo in tabella l’insieme di equazioni che risolvono analiticamente il problema del meato linearmente decrescente.

	2.3)
	
	gradiente della pressione

	2.13)
	
	pressione

	2.7)
	
	punto di massima pressione

	2.6)
	
	altezza di massima pressione

	2.17)
	
	velocità

	2.15)	
	
	sostentamento

	2.16)
	
	punto di applicazione del sostentamento

	2.24)
	
	coef. di attrito mediato



Nella seguente figura è invece riassunto il flusso delle deduzioni.

3. L’algoritmo risolutivo. Il programma che risolve questo quesito è costituito dalle seguenti unità:

· l’unità chiamante main_ese_sette, la quale si occupa di  risolvere la parte analitica, ovvero di calcolare  per via analitica; per il resto utilizza le subroutine di modulo;
· il modulo mod_ese_sette, il quale contiene le costanti fisiche e geometriche del problema, oltre alle seguenti subroutine:
· diagramma_p_primo, il quale si occupa di disegnare il grafico del gradiente della pressione;
· diagramma_p, il quale si occupa di disegnare il diagramma della pressione, tanto quella esatta che quella calcolata per via numerica;
· diagramma_v, che disegna i grafici delle velocità in funzione di , nei tre punti del meato di ascisse ;
· cavalieri che integra col metodo Cavalieri-Simpson le funzioni inviategli, ed è utilizzata per calcolare  a partire da , per calcolare  a partire da   e per calcolare  a partire da .

Segue il diagramma di flusso.
 (
invoca 
diagramma_p_primo
 
per disegnare il grafico del gradiente esatto 
diagramma_p_primo
Inizio
L’unità chiamante (
main_ese_sette
) calcola  
il gradiente della pressione esatto
la pressione esatta
il sostentamento esatto
il punto 
 esatto
invoca 
diagramma_p
 
per disegnare il grafico della pressione esatta 
diagramma_p
invoca 
cavalieri 
per calcolare la pressione approssimata 
cavalieri
invoca 
diagramma_p
 
per disegnare il grafico della pressione approssimata 
diagramma_p
invoca 
cavalieri 
per calcolare il sostentamento approssimato, lo mette in un array di cui ci interessa l’ultimo valore
cavalieri
)















	


















 (
invoca 
cavalieri 
per calcolare 
 approssimato, integrando la funzione 
 in 
; risolvendo poi l’equazione  
 otteniamo 
; utilizza 
 approssimata e 
 esatta
cavalieri
calcola la velocità nei punti 
 e la mette in tre array monodimensionali
invoca 
diagramma_v
 
per ciascuno dei tre array delle velocità
diagramma_v
calcola il massimo della pressione esatta e il massimo della pressione approssimata utilizzando MAXVA(array), funzione intrinseca del Fortran
scrive sullo schermo
la pressione massima esatta;
la pressione massima approssimata;
il sostentamento esatto
il suo punto di applicazione
il sostentamento approssimato
il suo punto di applicazione
il coefficiente di attrito mediato
calcola il coefficiente di attrito mediato con la formula analitica fornita dalla teoria
fine
)






































4. L’output. Riporto gli output numerici e grafici del programma. I primi sono riassunti nella tabella seguente.

	
	soluzione esatta
	soluzione numerica

	pressione massima
	385714.13 Pa
	385711.34 Pa

	sostentamento 
	94636.664 N/m
	94634.51 N/m

	punto di applicazione 
	0.23607785 m
	0.2360758 m

	coef. di attrito mediato
	0.0018491517



Il diagramma del gradiente della pressione è riportato nel seguito, insieme ai diagrammi della pressione calcolata in modo analitico e alla sua approssimazione ottenuta per integrazione numerica.

Riporto inoltre i diagrammi delle velocità in funzione di  in corrispondenza dei tre punti richiesti

4.1)	

segnalando che le tre altezze del meato ad essi relative sono rispettivamente

4.2)	
[image: ]

5. Secondo quesito, problema diretto. Abbiamo un cuscinetto Michell a pattini auto-orientabili per il quale si abbiano i seguenti dati



5.1)	

essendo il  il carico spingente,  la velocità di rotazione dell’albero  e  la viscosità (alla temperatura di funzionamento). Il significato degli altri parametri è spiegato attraverso la figura.
[image: ]
Si chiede dunque di risolvere il problema diretto ovvero di ricavare 

5.2)	le altezze  dei pattini;
5.3)	il coefficiente di attrito mediato .
5.4)	la pressione media.

Si precisa che il cuscinetto presenta otto pattini uguali.

6. Discussione analitica. Il cuscinetto Michell a pattini auto-orientabili si può considerare come una successione di meati ad altezza decrescente, tanti quanti sono i pattini. Dunque per ottenere la soluzione del problema si procede come segue.

· il punto di applicazione di ciascuna forza di sostentamento è noto, cadendo sul perno[footnoteRef:2]; dunque la  2.16 porge , come soluzione dell’equazione trascendente [2:  Infatti all’equilibrio il momento di ciascuna forza  deve essere nullo rispetto al perno stesso. ] 


6.1)	

· la forza di sostentamento offerta da ciascun pattino è nota e vale , sostituendola nella 2.15 ricavo  attraverso la

6.2)	

· da  e  si ricava ;
· si ricava il coefficiente di attrito mediato usando direttamente la 2.14.

7. L’algoritmo risolutivo. L’unica difficoltà risiede nella soluzione della equazione non lineare 6.1 per la quale si può adottare il metodo di Newton-Raphson. Per applicare questo metodo definisco la funzione

7.1)	

e ne calcolo la derivata in . Allora l’algoritmo di Newton-Raphson assume la forma

7.4)	

e un possibile innesco, dato il significato fisico di , potrebbe essere . Poiché la derivata risulta macchinosa da calcolare e da scrivere nel codice, si può approssimare con il rapporto incrementale, ponendo ad esempio

7.5)	

Nel codice ho richiesto di plottare la curva di , così da avere una verifica al risultato ottenuto col metodo di Newton-Raphson; la curva che si ottiene è riportata in figura, e dalla sua lettura è possibile avere una buona stima del valore di .
[image: ]












 


Il diagramma di flusso relativo al metodo di Newton-Raphson è indicato in figura. 
 (
resto <=0.001
falso
vero
la funzione 
psi 
calcola 
)































Descrivo ora il codice nel suo complesso. Esso è costituito dalle seguenti unità:

· l’unità chiamante main_ese_sette_due, la quale si occupa di  calcolare  per via analitica; calcola inoltre  attraverso il metodo di Newton-Raphson;
· il modulo mod_ese_sette_due, il quale contiene le costanti fisiche e geometriche del problema, oltre alle seguenti subroutine:
· diagramma_p_primo, il quale si occupa di disegnare il grafico del gradiente della pressione;
· diagramma_p, il quale si occupa di disegnare il diagramma della pressione, tanto quella esatta che quella calcolata per via numerica;
· diagramma_psi, che disegna il grafico della funzione psi la cui radice è ;
· cavalieri che integra col metodo Cavalieri-Simpson la pressione, calcolando così il sostentamento dovuto a un pattino, da confrontare con il carico assegnato, per verifica;
· psi, funzione che calcola il valore della funzione psi in corrispondenza del valore n inviatogli.

Il diagramma di flusso complessivo è indicato in figura.
 (
Inizio
invoca 
diagramma_p
 
per disegnare il grafico della pressione 
diagramma_p
invoca 
diagramma_p_primo
 
per disegnare il grafico del gradiente 
diagramma_p_primo
invoca 
cavalieri 
per calcolare il sostentamento approssimato, lo mette in un array di cui ci interessa l’ultimo valore
cavalieri
calcola il coefficiente caratteristico 
 cercando lo zero (con Newton-Raphson) della funzione 
7.1
.  
calcola 
 utilizzando la 
6.2
 
calcola 
L’unità chiamante (
main_ese_sette_due
) calcola  
il gradiente della pressione 
la pressione 
la pressione massima
la pressione media
il coefficiente di attrito mediato
invoca 
diagramma_psi
 
per disegnare il grafico della funzione 
psi
 
diagramma_psi
scrive sullo schermo
la pressione massima
la pressione media
il sostentamento dato
il sostentamento approssimato
il coefficiente di attrito mediato
fine
)










	






































Seguono i grafici della pressione e del gradiente per il singolo pattino:

Gli output numerici del programma sono riassunti nella tabella seguente:
 
	
	
	
	
	

	386532.94 
	226757.39
	16666.668
	3.119554
	0.007603118

	
	
	
	

	
	16666.5
	0.00036638076
	

	
	
	
	

	
	
	0.00011744652
	



8. Unità chiamante per il primo quesito. Riporto il codice in Fortran.
!main program per la settima esercitazione di meccanica apl.
!utilizza il modulo mod_ese_sette
!25/11/2012

PROGRAM main_ese_sette

USE DISLIN      		!libreria grafica
USE mod_ese_sette 		!modulo con costanti e procedure

!sezione dichiarativa

IMPLICIT NONE

!dichiaro alcune variabili utilizzate
REAL::x1=0.15    		!punto del meato dove si calcola v
REAL::x2			!idem
REAL::x3=0.35			!idem
REAL::h_x1, h_x2, h_x3
INTEGER::i,j 			!indici dei cicli
CHARACTER(len=10):: chiusura 	!serve per chiudere il programma
CHARACTER(len=25):: nome        !si usa per il titolo dei grafici
CHARACTER(len=15):: unita       !si usa per le unita' di misura dei grafici


!dichiaro gli array di tutte le funzioni coinvolte
REAL,DIMENSION(1:inter):: h	      !altezza del meato
REAL,DIMENSION(1:inter):: x           !array delle ascisse
REAL,DIMENSION(1:inter):: p_primo     !derivata della pressione
REAL,DIMENSION(1:inter):: p_esatta    !pressione esatta
REAL,DIMENSION(1:inter):: p_cavalieri !pressione ottenuta per via numerica
REAL,DIMENSION(1:inter):: integrando  !serve a calcolare xN-cavalieri
REAL,DIMENSION(1:inter):: primit      !idem


!dichiaro le variabili relative al sostentamento
REAL:: N_esatta                     !N esatta
REAL,DIMENSION(1:inter)::N_cavalieri!N ottenuta per via numerica
REAL::xN_esatto			    !punto di applicazione di N esatto
REAL::xN_cavalieri                  !punto di applicazione di N per via numerica
REAL::cost_N_esatta		    !costante moltiplicativa di N_esatta


!dichiaro le variabili relative alla pressione
REAL::p_max_esatta		    !la pressione esatta massima
REAL::p_max_cavalieri		    !la pressione numerica massima
REAL::cost_p_esatta		    !costante moltiplicativa di p_esatta
REAL::p_primo_x1 
REAL::p_primo_x2
REAL::p_primo_x3


!dichiaro le variabili relative all'asse z e alla velocita'
REAL,DIMENSION(1:inter2)::z                 !array delle z
REAL:: delta_z_x1, delta_z_x2, delta_z_x3   !incrementi
REAL,DIMENSION(1:inter2):: v_x1             !velocita' in x_1
REAL,DIMENSION(1:inter2):: v_x2             !velocita' in x_2
REAL,DIMENSION(1:inter2):: v_x3             !velocita' in x_3


!dichiaro le variabili relative al coefficiente di attrito
REAL::fm                        !coefficiente di attrito mediato
REAL::kn			!parte di fm
REAL::cost_fm			!idem


!altre variabili
REAL::delta_x 		    	!incremento dell ascisse
REAL::h_max                     !altezza in cui p è max
REAL::x_max                     !ascissa in cui p è max
REAL::n                     	!coefficente caratteristico


!sezione esecutiva


!alcune operazioni preliminari
delta_x = L/inter              !incremento dell ascisse
  h_max = (2.*h1*h2)/(h1+h2) !altezza in cui p e' max
  x_max = L*h1/(h1+h2)      !ascissa in cui p e' max
      n = h1/h2             !coefficente caratteristico


!dichiaro l'array delle ascisse
x(1) = 0.
ciclo_x: DO i=2,inter,1
x(i) = x(i-1) + delta_x
END DO ciclo_x


!dichiaro l'array delle z
z(1) = 0.
ciclo_z: DO i=2,inter2,1
z(i) = z(i-1) + delta_x
END DO ciclo_z



!calcolo l'altezza del meato
ciclo_h: DO i=1,inter,1
h(i) = h1 + (h2-h1)*x(i)/L
END DO ciclo_h


!calcolo la derivata della pressione
ciclo_p_primo: DO i=1,inter,1
p_primo(i) = (6*mu*V)*(h(i)-h_max)/(h(i)**3)
END DO ciclo_p_primo


!calcolo la pressione esatta
cost_p_esatta = (6*mu*V*L)/((n-1)*h2)
ciclo_p: DO i=1,inter,1
p_esatta(i) = cost_p_esatta*((1/h(i))-((h2/(h(i)**2))*(n/(n+1)))-(1/(h2*(n+1))))
END DO ciclo_p


!calcolo la pressione massima esatta
p_max_esatta = MAXVAL(p_esatta)


!calcolo il sostentamento esatto
cost_N_esatta = (6*mu*V*(L**2))/((h2**2)*((n-1)**2)*(n+1))
N_esatta = cost_N_esatta*(((n+1)*LOG(n))-2*(n-1))


!calcolo xN esatto
xN_esatto = L*((2*n*(n+2)*LOG(n))-(5*(n**2))+4*n+1)/(2*(n-1)*((n+1)*LOG(n)-2*(n-1)))



CALL BMPMOD (300,'inch','resolution') !fisso risoluzione dell'immagine in formato .bmp


!traccio il diagramma del gradiente della pressione
nome  = "gradiente della pressione"
unita = "\frac{Pa}{m}   "
CALL diagramma_p_primo (p_primo, x, nome, unita)


!traccio il diagramma della pressione esatta
nome = "pressione esatta         "
unita= "Pa             "
CALL diagramma_p (p_esatta, x, nome, unita)


!calcola la pressione col metodo cavalieri
CALL cavalieri (p_primo, p_cavalieri, p_esatta(1), p_esatta(2))


!traccio il diagramma della pressione secondo cavalieri
nome = "pressione Cavalieri      "
unita= "Pa             "
CALL diagramma_p (p_cavalieri, x, nome, unita)


!calcolo la pressione numerica massima
p_max_cavalieri = MAXVAL(p_cavalieri)


!calcola il sostentamento col metodo cavalieri
CALL cavalieri (p_esatta, N_cavalieri, 0., p_esatta(2)*delta_x/2)


!calcola il punto di applicazione col metodo cavalieri
ciclo_x_N: DO i=1,inter,1
integrando(i) = p_esatta(i)*x(i)
END DO ciclo_x_N
CALL cavalieri (integrando, primit, 0., p_esatta(2)*x(2)*x(2))
xN_cavalieri = primit(inter)/N_cavalieri(inter)


!calcolo la velocità nei punti richiesti
h_x1       = h1 + (h2-h1)*x1/L
h_x2       = h1 + (h2-h1)*xN_esatto/L
h_x3       = h1 + (h2-h1)*x3/L
p_primo_x1 = (6*mu*V)*(h_x1-h_max)/(h_x1**3)
p_primo_x2 = (6*mu*V)*(h_x2-h_max)/(h_x2**3)
p_primo_x3 = (6*mu*V)*(h_x3-h_max)/(h_x3**3)
delta_z_x1 = h_x1/inter2
delta_z_x2 = h_x2/inter2
delta_z_x3 = h_x3/inter2

z=0
ciclo_z_x1: DO j=2,inter2,1
z(j)=z(j-1)+delta_z_x1
END DO ciclo_z_x1
ciclo_v_x1: DO j=1,inter2,1
v_x1(j)=(p_primo_x1*(z(j)-h_x1)*z(j)/(2.*mu)) + (V*z(j)/h_x1)
END DO ciclo_v_x1
!plotto la velocità in x1
nome = "velocita' in x1=0.15m    "
unita= "m/s            "
CALL diagramma_v (v_x1,z ,h_x1 ,nome, unita)

z=0
ciclo_z_x2: DO j=2,inter2,1
z(j)=z(j-1)+delta_z_x2
END DO ciclo_z_x2
ciclo_v_x2: DO j=1,inter2,1
v_x2(j)=((1./(2.*mu))*p_primo_x2*(z(j)-h_x2)*z(j))+(V*z(j)/h_x2)
END DO ciclo_v_x2
!plotto la velocità in x2
nome = "velocita' in xN esatto   "
unita= "m/s            "
CALL diagramma_v (v_x2,z ,h_x2 ,nome ,unita)

z=0
ciclo_z_x3: DO j=2,inter2,1
z(j)=z(j-1)+delta_z_x3
END DO ciclo_z_x3
ciclo_v_x3: DO j=1,inter2,1
v_x3(j)=((1./(2.*mu))*p_primo_x3*(z(j)-h_x3)*z(j))+(V*z(j)/h_x3)
END DO ciclo_v_x3
!plotto la velocità in x3
nome = "velocita' in x3=0.35m    "
unita= "m/s            "
CALL diagramma_v (v_x3, z, h_x3, nome, unita)


!calcolo il coefficiente di attrito mediato
kn      = SQRT(8./3.)*(LOG(n)-((3*(n-1))/(2*(n+1))))/SQRT(LOG(n)-2*((n-1)/(n+1)))
cost_fm = SQRT(mu*V/N_esatta)
fm      = kn*cost_fm


!scrivo sullo schermo i dati calcolati
WRITE (*,*) " "
WRITE (*,*) "La pressione massima esatta vale", p_max_esatta, "Pa"
WRITE (*,*) " "
WRITE (*,*) "La pressione massima secondo Cavalieri vale", p_max_cavalieri, "Pa"
WRITE (*,*) " "
WRITE (*,*) "Il sostentamento esatto vale", N_esatta, "N/m"
WRITE (*,*) " "
WRITE (*,*) "Il suo punto di applicazione e'", xN_esatto, "m"
WRITE (*,*) " "
WRITE (*,*) "Il sostentamento secondo Cavalieri vale", N_cavalieri (inter), "N/m"
WRITE (*,*) " "
WRITE (*,*) "Il suo punto di applicazione e'", xN_cavalieri, "m"
WRITE (*,*) " "
WRITE (*,*) "Il coefficiente di attrito mediato e'", fm
WRITE (*,*) " "


WRITE (*,*) "le tre altezze sono", h_x1, h_x2, h_x3
WRITE (*,*) "l'altezze h-max e'", h_max
WRITE (*,*) "le tre velocità massime sono", v_x1(inter2), v_x2(inter2), v_x3(inter2)
WRITE (*,*) "variabilità della terza z", z(1), z(inter2)


!chiusura del programma
WRITE (*,*)"Per chiudere il programma premi una lettera qualunque."
WRITE (*,*)"Tutti i dati andranno persi."
READ  (*,*) chiusura


STOP

END PROGRAM main_ese_sette

9. Modulo per il primo quesito. Riporto il codice in Fortran.

!modulo per la settima esercitazione di meccanica apl.
!invocato dal programma main_ese_uno
!25/11/2012

MODULE mod_ese_sette

!sezione dichiarativa

IMPLICIT NONE

!fisso i valori delle costanti fisiche del problema

REAL, PARAMETER::    L  =   0.4             !lunghezza del meato in m
REAL, PARAMETER::    V  =   2.4             !velocita' della slitta in m/s
REAL, PARAMETER::    h1 =   0.00040         !altezza maggiore in m
REAL, PARAMETER::    h2 =   0.00016         !altezza minore in m
REAL, PARAMETER::    mu =   0.040           !viscosità in Ns/m^2

INTEGER, PARAMETER:: inter  = 1000          !lunghezza degli array
INTEGER, PARAMETER:: inter2 = 100           !lunghezza array z

!fisso il formato di output per i diagrammi

CHARACTER(len=4):: formato ='bmp '

!scrivo le subroutine 

CONTAINS

!-----------------------------------------------------------------------------------

SUBROUTINE diagramma_p_primo (funzione, x, nome, unita)

!traccia il diagramma della funzione passatagli dal main program

IMPLICIT NONE

!dichiaro gli argomenti fittizi

REAL, INTENT(IN), DIMENSION (1:inter):: x        !contiene i valori delle ascisse
REAL, INTENT(IN), DIMENSION (1:inter):: funzione !funzione da plottare
CHARACTER (len=25), INTENT(IN)::        nome     !titolo              !
CHARACTER (len=15), INTENT(IN)::        unita    !ascisse

!dichiaro le variabili locali

INTEGER:: i    !indice del ciclo

REAL::    max  !valore massimo della funzione
REAL::    min  !valore minimo della funzione

!sezione esecutiva

CALL METAFL (formato)  !indico il formato dell'output
CALL SCRMOD ('revers') !scritta nera su fondo bianco

CALL DISINI

CALL TEXMOD ('ON')     !chiedo di poter scrivere le formule in maniera grafica
CALL PAGERA            !traccio un bordo per il piano xy
CALL DUPLX             !font a doppio spessore

CALL AXSPOS (550,1800) !coordinate angolo basso sinistra
CALL AXSLEN (2200,1500)!lunghezza dei due assi in pixel

CALL NAME ('metri','x')       !nome delle ascisse
CALL NAME (unita,'y')       !nome delle ordinate

CALL TITLIN (nome,1) 	      !prima riga del titolo

max=MAXVAL(funzione)          !il massimo della funzione
min=MINVAL(funzione)          !il minimo  della funzione

CALL GRAF (0.,L,0.,0.1,min,max,min,1000000.)

CALL TITLE !stampa il titolo di cui sopra

CALL CURVE (x,funzione,inter) !plotto la funzione

CALL DASH   !tratteggio per gli assi coordinati
CALL XAXGIT !traccio la retta y=0
CALL YAxGIT !traccio la retta x=0

CALL DISFIN

RETURN

END SUBROUTINE diagramma_p_primo

!-----------------------------------------------------------------------------------

SUBROUTINE diagramma_p (funzione, x, nome, unita)

!traccia il diagramma della funzione passatagli dal main program

IMPLICIT NONE

!dichiaro gli argomenti fittizi

REAL, INTENT(IN), DIMENSION (1:inter):: x        !contiene i valori delle ascisse
REAL, INTENT(IN), DIMENSION (1:inter):: funzione !funzione da plottare
CHARACTER (len=25), INTENT(IN)::        nome     !titolo              !
CHARACTER (len=15), INTENT(IN)::        unita    !ascisse

!dichiaro le variabili locali

INTEGER:: i    !indice del ciclo

REAL::    max  !valore massimo della funzione
REAL::    min  !valore minimo della funzione

!sezione esecutiva

CALL METAFL (formato)  !indico il formato dell'output
CALL SCRMOD ('revers') !scritta nera su fondo bianco

CALL DISINI

CALL TEXMOD ('ON')     !chiedo di poter scrivere le formule in maniera grafica
CALL PAGERA            !traccio un bordo per il piano xy
CALL DUPLX             !font a doppio spessore

CALL AXSPOS (450,1800) !coordinate angolo basso sinistra
CALL AXSLEN (2200,1500)!lunghezza dei due assi in pixel

CALL NAME ('metri','x')       !nome delle ascisse
CALL NAME (unita,'y')       !nome delle ordinate

CALL TITLIN (nome,1) 	      !prima riga del titolo

max=MAXVAL(funzione)          !il massimo della funzione
min=MINVAL(funzione)          !il minimo  della funzione

CALL GRAF (0.,L,0.,0.1,min,max,min,100000.)

CALL TITLE !stampa il titolo di cui sopra

CALL CURVE (x,funzione,inter) !plotto la funzione

CALL DASH   !tratteggio per gli assi coordinati
CALL XAXGIT !traccio la retta y=0
CALL YAxGIT !traccio la retta x=0

CALL DISFIN

RETURN

END SUBROUTINE diagramma_p

!-----------------------------------------------------------------------------------


SUBROUTINE diagramma_v (funzione, z, h, nome, unita)

!traccia il diagramma della funzione passatagli dal main program

IMPLICIT NONE

!dichiaro gli argomenti fittizi

REAL, INTENT(IN), DIMENSION (1:inter2):: z        !contiene i valori delle ascisse
REAL, INTENT(IN), DIMENSION (1:inter2):: funzione !funzione da plottare
REAL, INTENT(IN):: h				  !altezza del meato
CHARACTER (len=25), INTENT(IN)::        nome      !titolo              
CHARACTER (len=15), INTENT(IN)::        unita     !ascisse

!dichiaro le variabili locali

INTEGER:: i    !indice del ciclo

REAL::    max  !valore massimo della funzione
REAL::    min  !valore minimo della funzione

!sezione esecutiva

CALL METAFL (formato)  !indico il formato dell'output
CALL SCRMOD ('revers') !scritta nera su fondo bianco

CALL DISINI

CALL TEXMOD ('ON')     !chiedo di poter scrivere le formule in maniera grafica
CALL PAGERA            !traccio un bordo per il piano xy
CALL DUPLX             !font a doppio spessore
CALL LABDIG (5,'y')    !5 cifre decimali per le ordinate

CALL AXSPOS (550,1800) !coordinate angolo basso sinistra
CALL AXSLEN (2200,1500)!lunghezza dei due assi in pixel

CALL NAME (unita,'x')       !nome delle ascisse
CALL NAME ('metri','y')       !nome delle ordinate

CALL TITLIN (nome,1) 	      !prima riga del titolo

max=MAXVAL(funzione)          !il massimo della funzione
min=MINVAL(funzione)          !il minimo  della funzione

CALL GRAF (0.,V,0.,0.5,0.,h,0.,0.00005)

CALL TITLE !stampa il titolo di cui sopra

CALL CURVE (funzione,z,inter2) !plotto la funzione

CALL DASH   !tratteggio per gli assi coordinati
CALL XAXGIT !traccio la retta y=0
CALL YAxGIT !traccio la retta x=0

CALL DISFIN

RETURN

END SUBROUTINE diagramma_v

!---------------------------------------------------------------------------------------

SUBROUTINE Cavalieri (funzione, primitiva, inizio_1, inizio_2)

!integra la funzione inviata con il metodo di Cavalieri

IMPLICIT NONE 

!dichiaro gli argomenti fittizi
REAL, INTENT(OUT),DIMENSION(1:inter)::primitiva !primitiva
REAL, INTENT(IN), DIMENSION(1:inter)::funzione  !integrando                              
REAL, INTENT(IN)::                    inizio_1   !il valore esatto di primitiva(1)
REAL, INTENT(IN)::                    inizio_2   !approssimazione di primitiva(2)

!dichiaro le variabili locali
INTEGER::i                              !indice del ciclo
REAL::delta_x = L/inter		    	!incremento dell ascisse    

!sezione esecutiva

primitiva(1)= inizio_1  !inizializza il primo valore
primitiva(2)= inizio_2  !inizializza il secondo valore

!calcolo la primitiva (elementi dispari dell'array)

ciclo_integra: DO i=1,inter-2,2

               primitiva(i+2)=primitiva(i)+((funzione(i+2)+funzione(i+1)*4+funzione(i))*(delta_x/3))

END DO ciclo_integra

!calcolo la primitiva (elementi pari dell'array)

ciclo_integra2: DO i=2,inter-2,2

               primitiva(i+2)=primitiva(i)+((funzione(i+2)+funzione(i+1)*4+funzione(i))*(delta_x/3))
               
END DO ciclo_integra2

RETURN

END SUBROUTINE Cavalieri

!-----------------------------------------------------------------------------------

END MODULE mod_ese_sette

10. Unità chiamante per il secondo quesito. Riporto il codice in Fortran.

!main program per la settima esercitazione (secondo quesito) di meccanica apl.
!utilizza il modulo mod_ese_sette_due
!26/11/2012

PROGRAM main_ese_sette_due

USE DISLIN      		!libreria grafica
USE mod_ese_sette_due 		!modulo con costanti e procedure




!sezione dichiarativa

IMPLICIT NONE

!dichiaro le variabili coinvolte nel calcolo di n
REAL:: n                                !il coefficiente caratteristico
REAL, DIMENSION(1:inter)::na            !l'array dei valori di n 
REAL:: resto                                           !si deve annullare
REAL:: h1,h2!                                  le due altezze


!dichiaro le variabili coinvolte col calcolo della pressione
REAL,DIMENSION(1:inter):: h	      !altezza del meato
REAL,DIMENSION(1:inter):: x           !array delle ascisse
REAL,DIMENSION(1:inter):: p_primo     !derivata della pressione
REAL,DIMENSION(1:inter):: p_esatta    !pressione esatta
REAL::p_max_esatta	 !la pressione esatta massima
REAL::p_max_cavalieri	 !la pressione numerica massima
REAL::cost_p_esatta                                                                                                !costante moltiplicativa di p_esatta
REAL::p_media	 !la pressione media

!dichiaro altre variabili utilizzate
CHARACTER(len=10):: chiusura       !serve per chiudere il programma
CHARACTER(len=25):: nome              !si usa per il titolo dei grafici
CHARACTER(len=15):: unita             !si usa per le unita' di misura dei grafici
INTEGER::i,j                          !indici di ciclo


!dichiaro le variabili relative al coefficiente di attrito
REAL::fm                        !coefficiente di attrito mediato
REAL::kn	!parte di fm
REAL::cost_fm			!idem


!dichiaro il sostentamento e la velocita'
REAL::N_dato
REAL, DIMENSION (1:inter)::N_cavalieri
REAL::V


!altre variabili
REAL::delta_x 		    	!incremento dell ascisse
REAL::h_max                     !altezza in cui p è max
REAL::x_max                     !ascissa in cui p è max


!dichiaro le variabili utilizzate per tracciare il diagramma di psi
REAL, DIMENSION (inter)::psia
REAL::u


!sezione esecutiva


!applico il metodo di newton-raphson per ricavare n
i=1					!valore iniziale dell'indice
na(i)=4.5 			!initial guess
ciclo: DO
na(i+1) = na(i) - ( psi(na(i))/((psi(na(i)+0.0001) - psi(na(i)))/0.0001))
resto = psi(na(i))  /  ((psi(na(i)+0.0001) - psi(na(i)))/0.0001) 
n=na(i+1)	
IF (resto<0.001) EXIT ciclo
i=i+1
END DO ciclo



!calcolo il sostentamento e la velocita'
N_dato = (P/8)/b
V = rm*omega


!calcolo h2 e h1
h2 = ( L/(n-1) )*SQRT(  6*mu*V*( (n+1)*LOG(n)-2*(n-1) )/(N_dato*(n+1))  )  
h1 = n*h2


!alcune operazioni preliminari
delta_x = L/inter              !incremento dell ascisse
  h_max = (2.*h1*h2)/(h1+h2) !altezza in cui p e' max
  x_max = L*h1/(h1+h2)      !ascissa in cui p e' max


!dichiaro l'array delle ascisse
x(1) = 0.
ciclo_x: DO i=2,inter,1
x(i) = x(i-1) + delta_x
END DO ciclo_x


!calcolo l'altezza del meato
ciclo_h: DO i=1,inter,1
h(i) = h1 + (h2-h1)*x(i)/L
END DO ciclo_h


!calcolo la derivata della pressione
ciclo_p_primo: DO i=1,inter,1
p_primo(i) = (6*mu*V)*(h(i)-h_max)/(h(i)**3)
END DO ciclo_p_primo


!calcolo la pressione esatta
cost_p_esatta = (6*mu*V*L)/((n-1)*h2)
ciclo_p: DO i=1,inter,1
p_esatta(i) = cost_p_esatta*((1/h(i))-((h2/(h(i)**2))*(n/(n+1)))-(1/(h2*(n+1))))
END DO ciclo_p


!calcolo la pressione massima esatta
p_max_esatta = MAXVAL(p_esatta)


!calcolo la pressione media 
p_media = N_dato/L


CALL BMPMOD (300,'inch','resolution') !fisso risoluzione dell'immagine in formato .bmp


!traccio il diagramma del gradiente della pressione
nome  = "gradiente della pressione"
unita = "\frac{Pa}{m}   "
CALL diagramma_p_primo (p_primo, x, nome, unita)


!traccio il diagramma della pressione esatta
nome = "pressione                "
unita= "Pa             "
CALL diagramma_p (p_esatta, x, nome, unita)



!calcolo il sostentamento integrando la pressione come verifica
CALL cavalieri (p_esatta, N_cavalieri, 0., p_esatta(2)*delta_x/2)


!calcolo il coefficiente di attrito mediato
kn      = SQRT(8./3.)*(LOG(n)-((3*(n-1))/(2*(n+1))))/SQRT(LOG(n)-2*((n-1)/(n+1)))
cost_fm = SQRT(mu*V/N_dato)
fm      = kn*cost_fm


!traccio il diagramma di psi
nome = "psi                      "
unita= "               "
na(1)=1
psia(1)=0
ciclo_psi: DO i=1, inter-1, 1
na(i+1)=na(i) + 0.01
u = na(i+1)
psia(i+1)= xN - L*((2*u*(u+2)*LOG(u))-(5*(u**2))+4*u+1)/(2*(u-1)*((u+1)*LOG(u)-2*(u-1)))
END DO ciclo_psi
CALL diagramma_psi (psia, na, nome, unita)


!scrivo sullo schermo i dati calcolati
WRITE (*,*) " "
WRITE (*,*) "La pressione massima vale", p_max_esatta, "Pa"
WRITE (*,*) " "
WRITE (*,*) "La pressione media vale", p_media, "Pa"
WRITE (*,*) " "
WRITE (*,*) "Il sostentamento assegnato vale", N_dato, "N/m"
WRITE (*,*) " "
WRITE (*,*) "Il sostentamento ricalcolato dalla pressione vale", N_cavalieri (inter), "N/m"
WRITE (*,*) " "
WRITE (*,*) "Il coefficiente di attrito mediato e'", fm
WRITE (*,*) " "
WRITE (*,*) "l'altezze h-max e'", h_max
WRITE (*,*) " "
WRITE (*,*) "l'altezze h1 e'", h1
WRITE (*,*) " "
WRITE (*,*) "l'altezze h2 e'", h2
WRITE (*,*) " "
WRITE (*,*) "il coefficente caratteristico del meato e'", n


!chiusura del programma
WRITE (*,*)"Per chiudere il programma premi una lettera qualunque."
WRITE (*,*)"Tutti i dati andranno persi."
READ  (*,*) chiusura


STOP

END PROGRAM main_ese_sette_due

11. Modulo per il secondo quesito. Riporto il codice in Fortran.

!modulo per la settima esercitazione di meccanica apl.
!invocato dal programma main_ese_sette_due
!26/11/2012

MODULE mod_ese_sette_due

!sezione dichiarativa

IMPLICIT NONE

!fisso i valori delle costanti fisiche del problema
REAL, PARAMETER::    L  =   0.0735          !lunghezza del meato in m
REAL, PARAMETER:: omega =  83.7758          !velocita' angolare dell'albero in rad/s
REAL, PARAMETER::    mu =   0.035           !viscosità in Ns/m^2
REAL, PARAMETER::    rm =   0.1		    !raggio medio del cuscinetto in m
REAL, PARAMETER::     b =   0.06            !larghezza della slitta in m
REAL, PARAMETER::    xN =   0.045           !punto di applicazione del carico in m
REAL, PARAMETER::     P =8000.0             !carico spingente complessivo

!fisso la lunghezza degli array utilizzati
INTEGER, PARAMETER:: inter  = 1000 

!fisso il formato di output per i diagrammi
CHARACTER(len=4):: formato ='bmp '


!scrivo le subroutine 
CONTAINS

!-----------------------------------------------------------------------------------

SUBROUTINE diagramma_p_primo (funzione, x, nome, unita)

!traccia il diagramma della funzione passatagli dal main program

IMPLICIT NONE

!dichiaro gli argomenti fittizi

REAL, INTENT(IN), DIMENSION (1:inter):: x        !contiene i valori delle ascisse
REAL, INTENT(IN), DIMENSION (1:inter):: funzione !funzione da plottare
CHARACTER (len=25), INTENT(IN)::        nome     !titolo              !
CHARACTER (len=15), INTENT(IN)::        unita    !ascisse

!dichiaro le variabili locali

INTEGER:: i    !indice del ciclo

REAL::    max  !valore massimo della funzione
REAL::    min  !valore minimo della funzione

!sezione esecutiva

CALL METAFL (formato)  !indico il formato dell'output
CALL SCRMOD ('revers') !scritta nera su fondo bianco

CALL DISINI

CALL TEXMOD ('ON')     !chiedo di poter scrivere le formule in maniera grafica
CALL PAGERA            !traccio un bordo per il piano xy
CALL DUPLX             !font a doppio spessore

CALL AXSPOS (550,1800) !coordinate angolo basso sinistra
CALL AXSLEN (2200,1500)!lunghezza dei due assi in pixel

CALL NAME ('metri','x')       !nome delle ascisse
CALL NAME (unita,'y')       !nome delle ordinate

CALL TITLIN (nome,1) 	      !prima riga del titolo

max=MAXVAL(funzione)          !il massimo della funzione
min=MINVAL(funzione)          !il minimo  della funzione

CALL GRAF (0.,L,0.,0.01,min,max,min,10000000.)

CALL TITLE !stampa il titolo di cui sopra

CALL CURVE (x,funzione,inter) !plotto la funzione

CALL DASH   !tratteggio per gli assi coordinati
CALL XAXGIT !traccio la retta y=0
CALL YAxGIT !traccio la retta x=0

CALL DISFIN

RETURN

END SUBROUTINE diagramma_p_primo

!-----------------------------------------------------------------------------------

SUBROUTINE diagramma_p (funzione, x, nome, unita)

!traccia il diagramma della funzione passatagli dal main program

IMPLICIT NONE

!dichiaro gli argomenti fittizi

REAL, INTENT(IN), DIMENSION (1:inter):: x        !contiene i valori delle ascisse
REAL, INTENT(IN), DIMENSION (1:inter):: funzione !funzione da plottare
CHARACTER (len=25), INTENT(IN)::        nome     !titolo              !
CHARACTER (len=15), INTENT(IN)::        unita    !ascisse

!dichiaro le variabili locali

INTEGER:: i    !indice del ciclo

REAL::    max  !valore massimo della funzione
REAL::    min  !valore minimo della funzione

!sezione esecutiva

CALL METAFL (formato)  !indico il formato dell'output
CALL SCRMOD ('revers') !scritta nera su fondo bianco

CALL DISINI

CALL TEXMOD ('ON')     !chiedo di poter scrivere le formule in maniera grafica
CALL PAGERA            !traccio un bordo per il piano xy
CALL DUPLX             !font a doppio spessore

CALL AXSPOS (450,1800) !coordinate angolo basso sinistra
CALL AXSLEN (2200,1500)!lunghezza dei due assi in pixel

CALL NAME ('metri','x')       !nome delle ascisse
CALL NAME (unita,'y')       !nome delle ordinate

CALL TITLIN (nome,1) 	      !prima riga del titolo

max=MAXVAL(funzione)          !il massimo della funzione
min=MINVAL(funzione)          !il minimo  della funzione

CALL GRAF (0.,L,0.,0.01,min,max,min,100000.)

CALL TITLE !stampa il titolo di cui sopra

CALL CURVE (x,funzione,inter) !plotto la funzione

CALL DASH   !tratteggio per gli assi coordinati
CALL XAXGIT !traccio la retta y=0
CALL YAxGIT !traccio la retta x=0

CALL DISFIN

RETURN

END SUBROUTINE diagramma_p


!---------------------------------------------------------------------------------------

SUBROUTINE diagramma_psi (funzione, x, nome, unita)

!traccia il diagramma della funzione passatagli dal main program

IMPLICIT NONE

!dichiaro gli argomenti fittizi

REAL, INTENT(IN), DIMENSION (1:inter):: x        !contiene i valori delle ascisse
REAL, INTENT(IN), DIMENSION (1:inter):: funzione !funzione da plottare
CHARACTER (len=25), INTENT(IN)::        nome     !titolo              !
CHARACTER (len=15), INTENT(IN)::        unita    !ascisse

!dichiaro le variabili locali

INTEGER:: i    !indice del ciclo

REAL::    max  !valore massimo della funzione
REAL::    min  !valore minimo della funzione

!sezione esecutiva

CALL METAFL (formato)  !indico il formato dell'output
CALL SCRMOD ('revers') !scritta nera su fondo bianco

CALL DISINI

CALL TEXMOD ('ON')     !chiedo di poter scrivere le formule in maniera grafica
CALL PAGERA            !traccio un bordo per il piano xy
CALL DUPLX             !font a doppio spessore
CALL LABDIG (5,'y')    !5 cifre decimali per le ordinate

CALL AXSPOS (450,1800) !coordinate angolo basso sinistra
CALL AXSLEN (2200,1500)!lunghezza dei due assi in pixel

CALL NAME ('h1/h2','x')         !nome delle ascisse
CALL NAME (unita,'y')       !nome delle ordinate

CALL TITLIN (nome,1) 	      !prima riga del titolo

max=MAXVAL(funzione(5:inter))          !il massimo della funzione
min=MINVAL(funzione)                   !il minimo  della funzione

CALL GRAF (2.,11.,2.,0.5,min,max,min,0.005)
CALL GRID (2,2)

CALL TITLE !stampa il titolo di cui sopra

CALL CURVE (x,funzione,inter) !plotto la funzione

CALL DASH   !tratteggio per gli assi coordinati
CALL XAXGIT !traccio la retta y=0
CALL YAxGIT !traccio la retta x=0

CALL DISFIN

RETURN

END SUBROUTINE diagramma_psi


!---------------------------------------------------------------------------------------

SUBROUTINE Cavalieri (funzione, primitiva, inizio_1, inizio_2)

!integra la funzione inviata con il metodo di Cavalieri

IMPLICIT NONE 

!dichiaro gli argomenti fittizi
REAL, INTENT(OUT),DIMENSION(1:inter)::primitiva !primitiva
REAL, INTENT(IN), DIMENSION(1:inter)::funzione  !integrando                              
REAL, INTENT(IN)::                    inizio_1   !il valore esatto di primitiva(1)
REAL, INTENT(IN)::                    inizio_2   !approssimazione di primitiva(2)

!dichiaro le variabili locali
INTEGER::i                              !indice del ciclo
REAL::delta_x = L/inter 	!incremento dell ascisse    

!sezione esecutiva

primitiva(1)= inizio_1  !inizializza il primo valore
primitiva(2)= inizio_2  !inizializza il secondo valore

!calcolo la primitiva (elementi dispari dell'array)

ciclo_integra: DO i=1,inter-2,2

               primitiva(i+2)=primitiva(i)+((funzione(i+2)+funzione(i+1)*4+funzione(i))*(delta_x/3))

END DO ciclo_integra

!calcolo la primitiva (elementi pari dell'array)

ciclo_integra2: DO i=2,inter-2,2

               primitiva(i+2)=primitiva(i)+((funzione(i+2)+funzione(i+1)*4+funzione(i))*(delta_x/3))
               
END DO ciclo_integra2

RETURN

END SUBROUTINE Cavalieri

!-----------------------------------------------------------------------------------

FUNCTION psi (u)

!calcola la funzione psi in u

!sezione dichiarativa

REAL:: psi
REAL:: u

!sezione esecutiva

psi = xN - L*((2*u*(u+2)*LOG(u))-(5*(u**2))+4*u+1)/(2*(u-1)*((u+1)*LOG(u)-2*(u-1)))

RETURN

END FUNCTION psi

!-----------------------------------------------------------------------------------


END MODULE mod_ese_sette_due

